

SPICE model memristora

Marinić, Ivan

Undergraduate thesis / Završni rad

2021

Degree Grantor / Ustanova koja je dodijelila akademski / stručni stupanj: **Josip Juraj Strossmayer University of Osijek, Faculty of Electrical Engineering, Computer Science and Information Technology Osijek / Sveučilište Josipa Jurja Strossmayera u Osijeku, Fakultet elektrotehnike, računarstva i informacijskih tehnologija Osijek**

Permanent link / Trajna poveznica: <https://um.nsk.hr/um:nbn:hr:200:695238>

Rights / Prava: [In copyright](#) / [Zaštićeno autorskim pravom.](#)

Download date / Datum preuzimanja: **2025-02-21**

Repository / Repozitorij:

[Faculty of Electrical Engineering, Computer Science and Information Technology Osijek](#)



**SVEUČILIŠTE JOSIPA JURJA STROSSMAYERA U OSIJEKU
FAKULTET ELEKTROTEHNIKE RAČUNARSTVA I
INFORMACIJSKIH TEHNOLOGIJA OSIJEK**

Sveučilišni preddiplomski studij elektrotehnike i informacijske tehnologije

SPICE MODEL MEMRISTORA

Završni rad

Ivan Marinić

Osijek, 2021.

**FERIT**FAKULTET ELEKTROTEHNIKE, RAČUNARSTVA
I INFORMACIJSKIH TEHNOLOGIJA OSIJEK**Obrazac Z1P - Obrazac za ocjenu završnog rada na preddiplomskom sveučilišnom studiju**

Osijek, 01.09.2021.

Odboru za završne i diplomske ispite

Prijedlog ocjene završnog rada na preddiplomskom sveučilišnom studiju

Ime i prezime studenta:	Ivan Marinić
Studij, smjer:	Preddiplomski sveučilišni studij Elektrotehnika i informacijska tehnologija
Mat. br. studenta, godina upisa:	4551, 24.07.2018.
OIB studenta:	41494935030
Mentor:	Izv.prof.dr.sc. Tomislav Matić
Sumentor:	
Sumentor iz tvrtke:	
Naslov završnog rada:	SPICE model memristora
Znanstvena grana rada:	Elektronika (zn. polje elektrotehnika)
Predložena ocjena završnog rada:	Vrlo dobar (4)
Kratko obrazloženje ocjene prema Kriterijima za ocjenjivanje završnih i diplomskih radova:	Primjena znanja stečenih na fakultetu: 2 bod/boda Postignuti rezultati u odnosu na složenost zadatka: 1 bod/boda Jasnoća pismenog izražavanja: 2 bod/boda Razina samostalnosti: 3 razina
Datum prijedloga ocjene mentora:	01.09.2021.
Datum potvrde ocjene Odbora:	08.09.2021.
Potpis mentora za predaju konačne verzije rada u Studentsku službu pri završetku studija:	Potpis:
	Datum:

**FERIT**FAKULTET ELEKTROTEHNIKE, RAČUNARSTVA
I INFORMACIJSKIH TEHNOLOGIJA **OSIJEK****IZJAVA O ORIGINALNOSTI RADA**

Osijek, 09.09.2021.

Ime i prezime studenta:

Ivan Marinić

Studij:

Preddiplomski sveučilišni studij Elektrotehnika i informacijska tehnologija

Mat. br. studenta, godina upisa:

4551, 24.07.2018.

Turnitin podudaranje [%]:

10

Ovom izjavom izjavljujem da je rad pod nazivom: **SPICE model memristora**

izrađen pod vodstvom mentora Izv.prof.dr.sc. Tomislav Matić

i sumentora

mog vlastiti rad i prema mom najboljem znanju ne sadrži prethodno objavljene ili neobjavljene pisane materijale drugih osoba, osim onih koji su izričito priznati navođenjem literature i drugih izvora informacija. Izjavljujem da je intelektualni sadržaj navedenog rada proizvod mog vlastitog rada, osim u onom dijelu za koji mi je bila potrebna pomoć mentora, sumentora i drugih osoba, a što je izričito navedeno u radu.

Potpis studenta:

Sadržaj

1. Uvod	1
1.1 Zadatak završnog rada.....	1
2. Memristor	2
2.1. Načelo rada.....	2
2.2.Strujno naponske karakteristike.....	4
2.3. Dinamička svojstva.....	6
3. SPICE model memristora	7
4. Rezultati simulacije.....	21
5. Zaključak.....	31
LITERATURA.....	32
PRILOZI.....	36
SAŽETAK	33
ABSTRACT.....	34
ŽIVOTOPIS	35

1. Uvod

U okviru ovog završnog rada izrađen je SPICE model memristora te su opisana njegova svojstva i provedena usporedba s ostalim memristorima. Simulacije su provedene u programskom paketu LTspice u kojem su provedene i usporedbe s dostupnim SPICE modelima memristora iz literature.

Memristor je naziv dobio po svojstvu memorije otpora, a prvi ga je definirao Leon Ong Chua kao element kruga s dva izvoda koji se ubraja u osnovne elemente uz otpornik, kondenzator i prigušnice, takozvani „element kruga koji nedostaje“.

Sam memristor predstavlja nelinearnu električnu komponentu koja povezuje električni naboj i magnetski tok. On omogućuje čuvanje podataka u vremenu bez potrebnog dodatnog vanjskog napajanja. Njegovo posebno svojstvo je da se njegov otpor može programirati i nakon toga ostaje pohranjen. Za razliku od električnog otpornika s fiksnim otporom, memristoru je otpor ovisan o naponu.

Memristorski materijal mora imati otpor koji se reverzibilno može mijenjati s naponom. Imaju vrlo jednostavnu strukturu, a najčešće se izrađuju od tankog filma od titanij-dioksida između dvije metalne elektrode.

Ovaj završni rad podijeljen je na pet poglavlja. U poglavlju „Memristor“ uveden je pojam i definicija memristora. Poglavlje „SPICE model memristora“ uvodi i pobliže opisuje teoretski model izrađenog SPICE modela memristora, dok poglavlje „Rezultati simulacije“ sadrži demonstraciju SPICE simulacija temeljenih na predloženom modelu. Na kraju je izveden zaključak.

1.1 Zadatak završnog rada

Opisati osnovna svojstva memristora. Napisati pregled stanja tehnike u području SPICE modeliranja memristora. Na temelju provedene analize stanja tehnike, izraditi SPICE model memristora za primjenu u simulaciji analognih sklopova. Analizirati rezultate simulacije za izrađeni model memristora.

2. Memristor

Memristor je električna komponenta čija se svojstva koriste za obradu i pohranu informacija. Ponaša se kao nelinearni otpor jer su njegove strujno naponske karakteristike nelinearne. Razlikuje se od otpora svojim dinamičkim svojstvima te posjeduje memoriju. On ograničava ili upravlja tokom električne struje te ima svojstvo pamćenja količine naboja u vremenu. Kako dođe do promjene napona na memristoru, kroz neko vrijeme prođe određena količina elektriciteta. Provodnost memristora ovisi o količini elektriciteta i promjeni napona ili struje koja prolazi kroz memristor. Ako dođe do isključenja napona napajanja, otpornost memristora ostaje ista. Dvopolna komponenta koja ima zatvorenu petlju histereze koja prolazi kroz ishodište koordinatnog sustava u strujno-naponskoj karakteristici pobuđena periodičnim izvorom, bez istosmjerne komponente, naziva se memristor.

2.1. Načelo rada

Inženjer Leon Chua pretpostavio je postojanje elementa koji povezuje naboj i tok. Navedenu komponentu definirao je kao memristor, eng. memory resistor.

Promatrajući četiri osnove veličine električnog kruga (struja i , napon u , magnetski tok φ , naboj q), mogu se definirati relacije koje povezuju naboj i tok kao vremenski promjenjive veličine:

$$q(t) = \int_{-\infty}^t i(\tau) d\tau = q_0 + \int_0^t i(\tau) d\tau, \quad (2-1)$$

$$\varphi(t) = \int_{-\infty}^t u(\tau) d\tau = \varphi_0 + \int_0^t v(\tau) d\tau, \quad (2-2)$$

gdje su q_0 i φ_0 početne vrijednosti u trenutku $t=0$.

Implicitne jednačbe koje opisuju osnovne, nelinearne elemente strujnog kruga su:

$$fR(u, i) = 0, \quad (2-3)$$

$$fL(\varphi, i) = 0, \quad (2-4)$$

$$fC(u, q) = 0. \quad (2-5)$$

U jednadžbama fR predstavlja funkciju otpora u ovisnosti od naponu i struji, fL funkciju induktiviteta u ovisnosti o magnetskom toku i struji i fC funkciju kapaciteta u ovisnosti o naponu i naboju.

Razlikujemo naponski (naboj) i strujno (tok) upravljane memristore.

Memristore kontrolirane nabojem opisujemo sljedećom jednadžbom:

$$v(t) = M(q(t)) * i(t), \quad (2-6)$$

gdje je:

$$M(q) = \frac{d\varphi(q)}{dq}. \quad (2-7)$$

U izrazu (2-7) $M(q)$ predstavlja memorijski otpor u vremenu, dq promjenu električnog naboja, a $d\varphi(q)$ promjenu magnetskog toka.

Memristore kontrolirane tokom opisujemo izrazima:

$$i(t) = G(\varphi(t))u(t), \quad (2-8)$$

$$G(\varphi) = \frac{dq(\varphi)}{d\varphi}. \quad (2-9)$$

U izrazu (2-9) $G(\varphi)$ predstavlja memvodljivost.

Za memristore se može primijeniti Ohmov zakon. Vodljivosti u trenutku t_0 ne ovise samo o tome trenutku nego i o prijašnjim trenucima što dokazuje svojstvo pamćenja. Ukoliko su definirani struja i napon u određenom trenutku t , memristor se ponaša kao vremenski promjenjiv otpornik. U slučaju kada je njegova karakteristika u $\varphi - q$ ravnini ravna linija, tada se ponaša kao vremenski nepromjenjiv otpornik.

Ukoliko dođe do velike promjene napona ili struje, može doći do velike promjene otpora te se samim time takvi elementi mogu upotrijebiti kao sklopke. Uz pretpostavku konstantnog napona,

memristoru koji mijenja vrijednost otpora sa R_{on} na R_{off} , naboj se mora promijeniti za ΔQ koji je razlika Q_{on} i Q_{off} kako bi se dobila energija potrebna za komutaciju E_{switch} .

- On- vrijednosti otvorene sklopke
- Off- vrijednosti zatvorene sklopke
- E_{switch} - energija potrebna za komutaciju

$$E_{switch} = U^2 \int_{T_{off}}^{T_{on}} \frac{dt}{M(q(t))}, \quad (2-10)$$

$$I(q) = \frac{dq}{dt}, \quad (2-11)$$

$$E_{switch} = U\Delta Q. \quad (2-12)$$

Proizlazi da je količina energije potrebne za jednu komutaciju jednaka umnošku napona na stezaljkama i ukupne promjene naboja memristora.




Slika 2.1. Simbol memristora [9]

2.2.Strujno naponske karakteristike


Memristor je elektronički element čije ulazno izlazne karakteristike nisu jednoznačno određene. Memristivnost predstavlja funkciju naboja ili toka. Vrijednost otpora samog memristora ovisi o čitavoj povijesti ulaza. Sve dok se mijenja ulaz, mijenja se i iznos memristivnosti. Ako

uklonimo signal, element održava, „pamti“, svoje stanje sve dok se ne promijeni ulaz. Slikom 2.2 se može prikazati statička ovisnost memristora o naboju i toku.




Slika 2.2. Ovisnost toka o naboju [6]

Kod memristora dolazi do pojave histereze po kojoj se razlikuje od otpornika kod kojeg je struja predstavljena kao jednoznačna funkcija napona. Kod memristora ista vrijednost napona može uzrokovati protjecanje dviju različitih struja, što uzrokuje petlju histereze.




Slika 2.3. Strujno naponska karakteristika [6]

2.3. Dinamička svojstva



Slika 2.4. Prikaz napona i struje u vremenu [6]

Dinamička svojstva memristora ispituju se na memristoru priključenom na sinusni izvor izmjeničnog napona. Sama karakteristika se mijenja promjenom frekvencije. Na slici 2.5. prikazana su četiri slučaja promjene frekvencije.



Slika 2.5. Strujno naponska karakteristika memristora [6]

3. SPICE model memristora

SPICE modeliranje i simulacija omogućava analizu elektroničkih krugova uz nužan kompromis između točnosti rezultata, brzine i pouzdanosti postupaka. Potrebno je provoditi pravilne postupke i postaviti pravilne postavke programskih mogućnosti za konkretne analize. U SPICE-u se koriste određeni algoritmi i ograničenja koja omogućavaju što točnije rezultate te sprječavanje konvergencije ukoliko dođe do nje. Implementacija samih kodova provodi se u raznim verzijama SPICE alata kao što su PSpice, LTspice, HSPICE itd. U ovom je radu korišten LTspice.

Modeliranje ponašanja neelektričnih veličina u SPICE-u je zasnovano na raznim analogijama. Na primjer modeliranje memristora može dovesti do odabira atipičnih vrijednosti parametara elemenata u zamjenskom električnom krugu. Kao rezultat dobiju se izrazito visoke ili niske vrijednosti napona i struja. Korisno je izbjegavati male vrijednosti otpora jer svaka pogreška takvih otpornika rezultira velikim strujnim pogreškama. Problemi s konvergencijom mogu nastati u sustavima s povratnim vezama s velikim dobicima petlje. Jedna od tehnika modeliranja koristi pasivne R , L i C elemente s negativnim vrijednostima. Ta metoda se ne preporučuje jer može izazvati nestabilno ponašanje samih elemenata.

Diskontinuirani modeli, eng. discontinuous models, rezultat su rada nekoliko memorijskih elemenata, na primjer memristorskih sustava s pragom ili multi-state memristora komutacijske memorije. Točno modeliranje ovih diskontinuiteta je strogo poželjno za pružanje visoke preciznosti modela. S druge strane, potencijalni izvor poteškoća predstavlja izvor numeričkih problema programa koje mogu smanjiti preciznost. Na taj problem se može utjecati na manjim sustavima modeliranjem diskontinuiteta karakteristike memorijskih elemenata u prvom koraku. U slučaju konvergencije ili nereálnih rezultata, može se koristiti tehnika naknadnog izravnivanja karakteristike. Na primjer funkcija IF se može modelirati tako da se varijable ne mijenjaju naglo kako bi se uklonili diskontinuiteti. Valni oblici signala mogu biti izvori diskontinuiteta čiji se impulsi modeliraju s realnim porastom/padom.

Modeli nekih analognih sklopova vrlo su osjetljivi na numeričke pogreške koje potječu od konačne preciznosti predstavljanja broja, što može biti posljedica određene operacije računskih algoritama. Model koji je izrađen može se različito ponašati u okruženjima različitih simulacijskih programa, čak i ako simulacije rade pod naizgled identičnim uvjetima. Izlazi iz simulacija mogu biti različiti od stvarnog ponašanja sustava koji se modeliraju. Pogreške su mogu uočiti, ali ne uvijek. Također je potrebno razlikovati izvor osjetljivosti modela koji može biti zbog modeliranog sklopa ili nepravilnog načina konstrukcije matematičkog modela. Modeli s velikim vrijednostima konstanti pokazuju veliku osjetljivost na numeričke pogreške koje djeluju na tranzijentne simulacije gdje se diferencijalne jednadžbe rješavaju numerički. Tipičan primjer osjetljivog kruga je idealni integrator koji je osnovni gradivni element idealnih memristora. Na bilo koji brojčani problem u proizvoljnim trenucima vremena tijekom integracije može utjecati algoritam privremene analize koji utječe na izračune u svim sljedećim trenucima.

U procesu modeliranja bitno je odabrati pravu metodu integracije i podesiti njezine parametre. S druge strane, osjetljivost modela na numeričke pogreške može se nepoželjno povećati nepravilnom konstrukcijom modela. Tipičan slučaj je nepravilno oduzimanje dva proporcionalna broja što rezultira velikom pogreškom u skraćivanju. Primjer za to je Joglekarova funkcija prozora za modeliranje nelinearnih svojstava memristora koji za parametar $p = 1$ mogu biti zapisani na 2 načina, prema [1]:

$$f(x) = 1 - (2x - 1)^2, \quad (3-1)$$

$$f(x) = 4x(1 - x). \quad (3-2)$$

Za memristore u graničnom stanju s maksimalnom memristancom, kada je x blizu 0, prvi model generira znatno veće pogreške te su za sve vrijednosti $x < 2,776 * 10^{-17}$ vrijednosti funkcije prozora smanjene na nulu. Za drugi model takvo ograničenje se pojavljuje ako je x manji od njegove minimalne vrijednosti za dvostruki tip. Takva osjetljivost modela na pogreške u skraćivanju može imati štetnu ulogu u svim uobičajenim modelima koji koriste funkcije prozora.

Pogreške u skraćivanju i njihovo nagomilavanje tijekom procesa integracije privremene analize može biti uzrok pogrešnih rezultata čak i za simulacije jednostavnih sklopova koji sadrže memeelemente. Razlog može biti nepravilan oblik jednadžbe stanja što rezultira velikom osjetljivošću na pogreške u skraćivanju. Ako se memorija memeelemenata približi graničnom stanju, SPICE može pogrešno procijeniti da je stanje već postignuto. Tada se vrijednost „zamrzne“ i dolazi do pogreške simulacije sustava. Za slučajeve kada je stanje elemenata daleko od graničnih stanja, simulacija je točna. Potrebno je odabrati prikladne varijable stanja.

Tijekom simulacije memristorskih sustava, izvedene formule ne bi trebale generirati nikakva dijeljenja malim brojevima niti druge numeričke pogreške (npr. svako oduzimanje proporcionalnih brojeva koje je osjetljivo na pogreške zaokruživanja). Ako je memorirana vrijednost uređaja blizu nule, poželjno je koristiti model zasnovan na naponu, a ne na izvoru struje. Ako je moguće podijeliti formulu za modeliranje memristance na fiksni i varijabilni dio, tada se fiksni dio može modelirati klasičnim fiksnim elementom, a preostali dio kontroliranim izvorom. Fiksni dio mora predstavljati pozitivnu vrijednost memristance. To pruža pouzdane modele preko Thevenina/Nortona bez ikakvih potencijalnih sukoba takvih veza idealnih izvora koji krše Kirchoffov zakon napona/struje. Programi obitelji SPICE omogućuju izravno modeliranje otpornika pomoću jednadžbi, koje mogu biti funkcija proizvoljnog napona ili struje, ili bilo koje druge systemske varijable kao što je vrijeme.

SPICE model je zasnovan na jednadžbama. U nastavku je prikazan model idealnog memristora. U memristoru koji kontrolira struja, V_M predstavlja napon na memristoru, a memristanca ovisi samo o naboju, naime, prema[1],

$$V_M = R(q(t))I, \quad (3-3)$$

s nabojem vezanim za struju putem izraza $I = dq / dt$. Koristi se model pretpostavke da se memorijski uređaj sastoji od dvije razine (niskog i visokog otpora) s pomičnim granicama. Ukupna memristanca može se zapisati kao zbroj otpora dviju razina

$$R(x) = R_{on} * x + R_{off} (1 - x). \quad (3-4)$$

Ovdje je $x \in [0, 1]$ i parametrizira položaj granice, R_{on} i R_{off} su granične vrijednosti memristance. Kretanje x može se napisati pomoću funkcije prozora $W(x)$ kao:

$$R \frac{dx}{dt} = kW(x)I, \quad (3-5)$$

gdje je k konstantan, a $W(x)$ je definiran sljedećom relacijom:

$$W(x) = 1 - (2x - 1)^{2p}, \quad (3-6)$$

gdje je p pozitivan cjelobrojni broj.

Ovaj model uzima u obzir granične vrijednosti. Ne uključuje prag uključivanja, nije stabilan protiv fluktuacija i pokazuje prekomjerno odgođeno prebacivanje. Može se integrirati proizvoljna funkcija $W(x)$ i na taj način se memristanca izraziti kao funkcija q . Na primjer, ako je $W(x)$ dan jednadžbom (3-6) uz $p = 1$, onda dolazimo do

$$\frac{1}{4} \ln \frac{x}{1-x} = k(q(t) + q_o), \quad (3-7)$$


gdje je q_o konstantan.

$$R(q(t)) = R_{off} + \frac{R_{on} - R_{off}}{ae^{-4k(q(t)+q_o)} + 1}, \quad (3-8)$$

$$a = \frac{R_{ini} - R_{on}}{R_{off} - R_{ini}}. \quad (3-9)$$

Prikazane jednadžbe prikazuju pouzdan model za SPICE simulaciju. Memristanca je izvedena kao funkcija varijable stanja q , stoga jednadžba stanja nije osjetljiva na pogreške u skraćivanju. U SPICE-u se punjenje kondenzatora C_{int} može dobiti integriranjem priključne struje I . Zatim je naboj izražen u C jednak naponu čvora Q izraženom u V. Memorijski port se može modelirati kao serijski spoj fiksnog otpora R_{off} i kontroliranog izvora napona. Za modeliranje velikih sklopova,


koji mogu biti skloni problemima konvergencije, koristi se Nortonov ekvivalent prema slici 3.1. b).



Slika 3.1. Moguće implementacije idealnog memristora [1]

Memorijski port se može modelirati putem izvora napona sa serijskim otpornikom (a) ili preko ekvivalentnog izvora struje s paralelnim otpornikom (b). Ispravnost rezultata može se procijeniti pomoću valnog oblika naboja, koji mora biti periodičan bez ikakvih početnih prijelaznih pojava. Za LTspice poželjno je koristiti Gear integraciju koja vodi do najboljih rezultata.

Memristor se primjenjuje u analognim sklopovima koji se najčešće baziraju na jednostavnom upravljačkom sklopu sastavljenom od izvora napajanja, memristora te dva FET tranzistora kao što je prikazano ka slici 3.3. FET-ovi su kontrolirani s dva signala U_{pp} i U_{pn} kojima se programira otpor memristora R_M . Prikazani napon U_{pr} predstavlja programirani napon koji bi trebao premašiti napon praga memristora.



Slika 3.2. Osnovni upravljački sklop memristora [8]

Modeli memristora su: -linearni model

-nelinearni model

-model prilagodljivog memristora (TEAM)

-MACRO model

Linearni model sadrži linearni odnos između varijabli stanja i napona/struje. Ne može se koristiti pri većim naponima. Predviđa obrnuti odnos između vremena uključivanja i primjenjenog napona.

$$V_o \propto \frac{1}{T_o}. \quad (3-10)$$

Nelinearni model je najtočniji model koji predviđa promjenjivo i statičko i dinamičko ponašanje memristora. Struja $i(t)$ se sastoji od dvije komponente, $\sinh()$ koja je posljedica tunelskog učinka i $\exp()$ koja je posljedica ponašanja PN spoja.

$$\frac{dW(t)}{dt} = a * V^m(t) * f(w), \quad (3-11)$$

$$i(t) = W^n(t)\beta \sinh(\alpha V(t)) + \chi[\exp(\gamma V(t)) - 1]. \quad (3-12)$$

TEAM model je jednostavniji i općenitiji model koji prikazuje nelinearni model memristora radi poboljšanja računalne učinkovitosti. Ispod određenog praga varijabla stanja mu se ne mijenja. Postoji polinomska ovisnost između struje memristora i varijable stanja.

$$\frac{dx(t)}{dt} = \begin{cases} k_{\text{off}} \cdot \left(\frac{i(t)}{i_{\text{off}}} - 1 \right)^{\alpha_{\text{off}}} \cdot f_{\text{off}}(x) & 0 < i_{\text{off}} < i, \\ k_{\text{on}} \cdot \left(\frac{i(t)}{i_{\text{on}}} - 1 \right)^{\alpha_{\text{on}}} \cdot f_{\text{on}}(x) & i < i_{\text{on}} < 0, \end{cases} \quad (3-13)$$

gdje su k_{off} i k_{on} konstante, a i_{off} i i_{on} pragovi struje.

TEAM je precizan, s prosječnom pogreškom od 0,2% i zadovoljava uvjete konvergencije i računalnu učinkovitost.

MACRO model predstavlja model sastavljen od kruga koji se sastoji od različitih diskretnih električnih komponenti koje zajedno daju karakteristike poput memristora.

Funkcije prozora koriste se u modelima za sprječavanje izlaska varijable stanja iz granica $[0, D]$, kao i za dodavanje više nelinearnosti u blizini granica. Ovisno o funkciji, u obzir se mogu uzimati rubni uvjeti na gornjoj ili donjoj elektrodi, ali ne nužno. Linearnost zanošenja na granicama može biti linearna i nelinearna. Pri niskim naponima, nelinearni model izvodi isto što i linearni model, oba s istim funkcijama prozora. Funkcije prozora mogu pružiti skalabilnost. Ovisno o fleksibilnosti funkcije, ona može pružiti kontrolne parametre za nekoliko svojstava finog podešavanja.

Nekoliko funkcija prozora:

-Strukovljeva

$$F(\omega) = \frac{\omega(1 - \omega)}{D^2}, \quad (3-14)$$

gdje je ω varijabla stanja, D je duljina dijela niske vodljivosti.

-Benderlijeva

$$F(\omega) = \frac{\omega(D - W)}{D^2}, \quad (3-15)$$

gdje je ω varijabla stanja, D je duljina dijela niske vodljivosti, W je duljina dijela s visokom vodljivošću.

-Joglekar

$$F(\omega) = 1 - (2\omega - 1)^{2p}, \quad (3-16)$$

gdje je ω varijabla stanja, p je parametar koji kontrolira iznos nelinearnosti.

-Biolek

$$F(\omega) = 1 - (\omega - \text{sign}(-i))^{2p}, \quad (3-17)$$

gdje je ω varijabla stanja, p je parametar koji kontrolira iznos nelinearnosti, a i je struja.

-Prodomakis

$$F(\omega) = j \cdot (1 - [(\omega - 0.5)^2 + 0.75]^p), \quad (3-18)$$

gdje je ω varijabla stanja, p je parametar koji kontrolira iznos nelinearnosti, j je faktor skalabilnosti.

-Jinxiang


$$F(\omega) = j \cdot (1 - [a(\omega - \text{stp}(-i))^{2p} + (1 - a)]^p), \quad (3-19)$$

gdje je ω varijabla stanja, p je parametar koji kontrolira iznos nelinearnosti, j je faktor skalabilnosti, a kontrolira razlučivost: $0 < a < 1$.

Izrada samog SPICE modela memristora kreće od analiziranja podataka o fizičkom i matematičkom modelu memristora. Fizički model memristora sastoji se od dvoslojnog tankog filma TiO_2 , veličine $D \approx 10\text{nm}$, između platinskih kontakata. Jedan od slojeva se ponaša kao poluvodič i predstavlja dopirano područje, dok se drugi ponaša kao izolacija i predstavlja nedopirano područje. Širina poluvodiča w je modulirana ovisno o količini prolaska električnog naboja kroz memristor. Ukupan otpor memristora R_{MEM} zbroj je otpora dopiranog i nedopiranog područja,

$$R_{MEM}(x) = R_{ON}x + R_{OFF}(1 - x), \quad (3-20)$$

gdje je $x = \frac{w}{D} \in (0,1)$ širina dopiranog područja koja utječe na duljinu D sloja TiO_2 , a granice su R_{OFF} i R_{ON} vrijednosti otpora za $w = 0$ i $w = D$.



Slika 3.3. Fizički model memristora [4]

Ohmov zakon primjenjiv je između napona i struje memristora na način:

$$v(t) = R_{MEM}(w)i(t). \quad (3-21)$$

Brzina kretanja granice između dopiranog i nedopiranog područja ovisi o otporu dopiranog područja, prolazne struje i drugih čimbenika prema jednadžbi stanja:


$$\frac{dx}{dt} = ki(t)f(x), k = \frac{\mu_v R_{ON}}{D^2}, \quad (3-22)$$

gdje je $\mu_v \approx 10^{-14} \text{ m}^2\text{s}^{-1}\text{V}^{-1}$ i predstavlja takozvanu mobilnost legura. Mali naponi mogu proizvesti ogromno električno polje, koje sekundarno stvara značajne nelinearnosti u ionskom transportu. Nelinearnosti se pojavljuju kod tankih rubova filma, pri čemu se brzina granice između dopiranog i nedopiranog područja postupno smanji na nulu. Ta pojava se naziva nelinearnim zanošenjem legure i može biti modelirana takozvanom funkcijom prozora oblika:


$$f(x) = 1 - (2x - 1)^{2p}, \quad (3-23)$$

gdje je p pozitivan cijeli broj. Oblik funkcije $f(x)$ osigurava nultu brzinu x-koordinata.


Jednadžba memristora može se modelirati pomoću fizičkog modela na slici 3.4. Memorijski učinak memristora modeliran je putem integratora upravljano povratnom vezom. S obzirom na ograničavanje graničnih uvjeta, pohranjuje učinke prolaska struje i kontrolira otpor memristora mijenjanjem graničnog položaja. Nelinearni pomak i utjecaj rubnih uvjeta modelirani su pomoću povratne veze putem nelinearne funkcije prozora $f(x)$ (slika 3.4.).



Slika 3.4. Funkcija prozora $f(x)$ za različite vrijednosti p [4]



Slika 3.5. Blok dijagram modela memristora [4]



Slika 3.6. Struktura SPICE modela [4]

Struktura SPICE modela prikazana je na slici 3.6. Odnos između napona memristora i struje je modeliran na temelju jednadžbe:

$$R_{MEM}(x) = R_{OFF} - x\Delta R, \Delta R = R_{OFF} - R_{ON}. \quad (3-24)$$


Na slici 3.7., jednadžba 3-14 odgovara R_{OFF} otporniku u seriji s izvorom napona E čiji je napon kontroliran prema formuli „ $x\Delta R$ “. Normalizirana širina x dopiranog sloja modelirana je pomoću napona $V(x)$ na kondenzatoru Cx , koji služi kao integrator veličina na desnoj strani jednadžbe stanja 3-12. Početno stanje normalizirane širine legiranog sloja x_0 , koja je modelirana kao početni napon kondenzatora, određeno je početnim otporom R_{INIT} memristora prema formuli:

$$x_0 = \frac{R_{OFF} - R_{INTT}}{\Delta R}. \quad (3-25)$$

Model je implementiran kao SPICE pod mreža s parametrima koji mogu prenijeti sljedeće argumente: početni otpor R_{INIT} , R_{OFF} i R_{ON} otpor, širinu taknog filma D , mobilnost μv i eksponent p funkcije prozora. Koristi se model funkcije prozora prema Joglekaru (3-16), uključujući uvoz eksperimentalno prikupljenih podataka. Model SPICE se također nadopunjuje izravnim proračunom integralnih veličina koje definiraju memristor, tj. vremenskim integralom električnog napona (tok eng. flux) i električne struje (naboj eng. charge). Ove vrijednosti pripadaju rezultatima SPICE analiza dostupnih kao naponi interno upravljanih izvora E_{flux} i E_{charge} .

Model memristora prikazan je na slici 3.7. i primjenjuje se za približnu emulaciju ponašanja modeliranog memristora. Oznaka gornje elektrode memristora je 'te', a donje elektrode 'be'. Memristorski ekvivalent se napaja izvorom sinusnog napona. Modelirana je memristorska struja izvorom struje GI , koji je preuzet iz standardne LTSPICE biblioteke. Struja ovog ovisnog izvora je memristorska struja koja upravlja naponom memristora, a teče kroz izvor napona VI . Unutarnji otpor izvora napona napajanja izražava se otpornikom RI iz standardne biblioteke LTSPICE. Dodatna elektroda za mjerenje memristorske varijable x označena je s Y . Vremenska derivacija varijable stanja dx/dt je proporcionalna struji memristora. Izražava se naponski ovisnim, upravljanim strujnim izvorom GY preuzetim iz biblioteke LTSPICE. Kondenzator CI odabran iz standardne biblioteke je spojen paralelno na izvor GY . Ovaj kondenzator se koristi za integriranje struje proporcionalne varijabli stanja memristora dx/dt . Pad napona na kondenzatoru CI je

ekvivalentan memristorskoj varijabli stanja x . Dodatni otpornik $R2$ ima vrijednost $1\text{ G}\Omega$ te ima svrhu sprječavanja problema konvergencije tijekom simulacija. Odgovarajući LTSPICE kod memristorskog modela IM za daljni opis prikazan je u prilogu.




Slika 3.7. Opća shema memristora K1 u LTSpice-u

Kod prikazan u prilogu napisan je prema jednadžbama 3-11 i 3-12 koje opisuju model memristora IM. Prvi red prikazanog koda definira memristorski podkrug IM, njegove elektrode „te“ i „be“ i dodatnu elektrodu Y za mjerenje varijable stanja. Parametri memristorskog modela IM su ON otpor R_{on} , OFF otpor R_{off} , konstanta k , b i cc parametri podešavanja, $uthr1$ negativni prag aktivacije, $uthr2$ pozitivni prag aktivacije i kapacitet integrirajućeg kondenzatora C s njihovim vrijednostima prikazanim u drugom retku. Spojevi elemenata $R2$ i $C1$ između elektroda Y i donje elektrode „be“ i početnog napona kondenzatora $C1$ su prikazani u retcima 3 i 4 te služe sprječavanju problema konvergencije. Peti red koda predstavlja ovisni izvor struje G_y na koji se primjenjuje upravljački napon između gornje elektrode „te“ i donje elektrode „be“. Struja izvora G_y proporcionalna je vremenskoj varijabli stanja, a dominantno teče kroz kondenzator $C1$. Kako je napon kondenzatora $C1$ proporcionalan vremenskom integralu njegove struje, potencijal V_Y elektrode Y je proporcionalan varijabli stanja memristora x . Šesti red koda odgovara diferencijalnoj jednadžbi stanja. Varijabla stanja ograničena je u intervalu $(0,1)$. Ovisni izvor struje opisan u 6. retku predstavlja struju memristora ovisnu o primijenjenom naponu $V1$ i varijabli stanja memristora x . Funkcija $stpp$ predstavlja stepenastu logičku funkciju za

generiranje modela biblioteke memristora za izbjegavanje problema konvergencije. Posljednji redak zaključuje LTSpice kod.

Za modeliranje modela memristora potrebno je opisani model pretvoriti u element biblioteke LTSpice. Tako se dobiva model memristora od titanijevog dioksida s Joglekarovom prozorskom funkcijom ovisnom o naponu primjenjiv za simulaciju u LTSpice-u. Model realno prikazuje nelinearni pomak koji raste s povećanjem primjenjenog napona. Ako se primijeni vrijednost napona niža od napona praga, varijabla stanja ovog modela se ne mijenja, a memristor se ponaša kao linearni otpornik. Dodatno smo primijenili funkciju *stpp* za izbjegavanje problema konvergencije te izveli i dodatnu elektrodu *Y* za mjerenje varijable stanja. Svojstva izvedenog modela su: niska srednja vrijednost radne frekvencije, niska razina signala, visoka nelinearnost i točnost i primijenjeni naponi pragova. Model se može koristiti u analizi analognih i digitalnih sustava te u neuronskim mrežama.




Slika 3.8. Prikaz modeliranog simbola memristora u LTSpice-u

4. Rezultati simulacije

LTspice omogućuje simulaciju memristorskih sklopova. To je besplatan program iz obitelji SPICE koji omogućuje kreiranje shema, simulaciju i prikaz samih signala i valnih oblika.

Program sadrži sve pasivne i aktivne električne komponente te omogućuje rješavanje problema i simulacija različitih sustava i logičkih krugova. Koristimo modele memristora s njihovim pripadajućim kodovima.

Makromodel memristora u LTSpice okruženju nastao je na temelju matematičkog modela HP laboratorija. Simbol je kreiran i testiran u krugu prikazanom na slici 3.3 koji se sastoji od izvora sinusnog napona s amplitudom 1,2V i frekvencijom 1HZ, uzimajući u obzir fazu od 0°. Početne vrijednosti su $R_{ON} = 100\Omega$, $R_{OFF} = 16k\Omega$, $R_{INIT} = 11k\Omega$, $D = 10nm$ i $\mu V = 10f$ za $T = 1s$.



Slika 4.1. LTspice makromodel

Prva primijenjena funkcija prozora bila je funkcija Strukov prozora. Simulacija je na temelju prijelazne analize provedenog makromodela za $D = 10nm$. Struja memristora varira oko $150 \mu A$ za primjenjeni napon od 1,2V. Primjenjujući Ohmov zakon, ukupni granični otpor za ovaj model memristora R_{MEM} iznosi oko $8 k\Omega$. Primjenom napona, memristor daje samo male

varijacije memristance što pokazuje da se ne koristi cijeli raspon promjene otpora memristora. Ovaj model nema fleksibilnost za kontrolu nelinearnosti na granicama.

Sljedeća funkcija prozora je Joglekar. Kao što je već spomenuto u poglavlju 3, kako se vrijednost varijable stanja p povećava, tako je i nelinearnost veća. Prikazani su tok struje i napona kroz vrijeme na slici 4.6. Struja varira oko $350 \mu A$ za primijenjeni napon od $0.9V$. Granični otpor R_{MEM} iznosi $2,5 k\Omega$. Strujno naponska karakteristika dokazuje da ovaj model doseže veću širinu raspona otpora memristora. Povećanjem parametra p petlja se povećava, odnosno postaje šira.

Dalje je promatrana Biolek funkcija prozora. Postignuta je struja oko $2,4mA$ uz parametar $p=7$ zadan u SPICE kodu. Otpor doseže vrijednost $1k\Omega$. Strujno naponska karakteristika slična je Prodromakisovoj te pokazuje veću širinu raspona otpora koji varira od $100k\Omega$ do $1k\Omega$. Povećanjem parametra p povećava se širina raspona.

Sljedeći ispitani model je Yakopcic. Postiže maksimalnu struju oko $9mA$ uz napon amplitude $1V$ i frekvencije $10Hz$. Otpor dostiže veliki raspon vrijednosti. Uspoređujući s ostalim modelima, varijabla stanja ima linearnije kretanje te ima definirane napone praga.


Pino model memristora sličan je Yakopcic modelu. Ima manji raspon vrijednosti otpora te pokazuje linearnost uz definirane napone praga. Testiran je na naponu amplitude $0.9V$ i frekvencije $10Hz$. Strujno naponska karakteristika prikazana je na slici 4.16.

Simulacija Michigan modela memristora prikazana je na slici 4.17. Primijenjen je napon amplitude $1V$ i frekvencije $1Hz$. Postoji preklapanje u petlji histereze uzorkovano difuzijom iona.


Knowm model je sličan Yakopcic modelu uz manji raspon vrijednosti otpora te radi na većoj frekvenciji. Model ima definirane vrijednosti praga pri kojima memristor radi odnosno ne radi.

Posljednji testirani model je modelirani IM memristor. Iz analize simulacije vidi se nelinearnost izvedenog memristora. Usporedno s ostalim, strujno naponska karakteristika najbližnja je Strukov modelu.


Prikaz svih simulacija nalazi se u nastavku.




Slika 4.2. Shematski prikaz strujnog kruga sa Strukov memristorom




Slika 4.3. Struja i napon Strukov memristora kroz vrijeme




Slika 4.4. Strujno naponska karakteristika Strukov memristora




Slika 4.5. Shematski prikaz strujnog kruga s Joglekar memristorom




Slika 4.6. Struja i napon Joglekar memristora kroz vrijeme




Slika 4.7. Strujno naponska karakteristika Joglekar memristora




Slika 4.8. Shematski prikaz strujnog kruga s Biolek memristorom




Slika 4.9. Struja i napon Biolek memristora kroz vrijeme




Slika 4.10. Strujno naponska karakteristika Biolek memristora




Slika 4.11. Shematski prikaz strujnog kruga s Yakopcic memristorom




Slika 4.12. Struja i napon Yakopcic memristora kroz vrijeme




Slika 4.13. Strujno naponska karakteristika Yakopcic memristora




Slika 4.14. Shematski prikaz strujnog kruga s Pino memristorom




Slika 4.15. Struja i napon Pino memristora kroz vrijeme




Slika 4.16. Strujno naponska karakteristika Pino memristora




Slika 4.17. Shematski prikaz strujnog kruga s Michigan memristorom




Slika 4.18. Struja i napon Michigan memristora kroz vrijeme




Slika 4.19. Strujno naponska karakteristika Michigan memristora




Slika 4.20. Shematski prikaz strujnog kruga s Knowm memristorom




Slika 4.21. Struja i napon Michigan memristora kroz vrijeme




Slika 4.22. Strujno naponska karakteristika Knowm memristora



Slika 4.23. Shematski prikaz strujnog kruga s vlastitim IM modeliranim memristorom



Slika 4.24. Struja i napon vlastitog IM memristora kroz vrijeme



Slika 4.25. Strujno naponska karakteristika vlastitog IM memristora

5. Zaključak

Memristor predstavlja veliko otkriće u elektronici. Proučavanjem svojstava i njegove primjene može se uočiti njegova važnost danas. Današnja tehnologija napreduje i temelji se na komponentama koje imaju sposobnost pamćenja kao što ima memristor. Karakteriziraju ga specifična statička i dinamička svojstva po kojima se razlikuje od običnog otpornika.

Iz proučenih dobivenih strujno naponskih karakteristika dostupnih SPICE modela memristora moguće je uočiti funkciju petlje histereze. Petlja histereza se mijenja s ozbirom na frekvenciju i model memristora. Simulacije dostupnih SPICE modela memristora provedene su u programu LTspice. Prikazane su petlje histereze različitih modela te je izrađen vlastiti SPICE model memristora koji je također simuliran. Memristor treba biti modeliran na temelju dostupnih parametara tehnologije u kojoj se izrađuje kako bi se mogao simulirati i analizirati njegov rad ciljem daljnje izrade krugova temeljenih na memristoru.

LTspice omogućuje jednostavnu simulaciju električnih krugova. Memristor omogućuje jednostavno reprogramiranje, mogućnost obrade podataka te neovisnost o izvoru napajanja.

LITERATURA

- [1] D. Biolek, M. Di Ventra, Y.V. Pershin, „Reliable SPICE Simulations of Memristors, Memcapacitors and Meminductors“, 2010.
- [2] C. Yakopcic, T. M. Taha, G. Subramanyam, R. E. Pino and S. Rogers, "A Memristor Device Model," in IEEE Electron Device Letters, vol. 32, no. 10, pp. 1436-1438, Oct. 2011, doi: 10.1109/LED.2011.2163292.
- [3] Roberto Marani, Gennaro Gelao, Anna Gina Perri „A review on memristor applications“, 2015.
- [4] Biolek, Zdenek & Biolek, Dalibor & V, Biolkova. „SPICE Model of Memristor with Nonlinear Dopant Drift. Radioengineering.“, 2009.
- [5] Dongale, Dr. T. & Mohite, S. & Bagade, A.A. & Gaikwad, Pawan & Patil, P. & Kamat, Rajanish & Rajpure, Keshav. „Development of Ag/WO3/ITO Thin Film Memristor Using Spray Pyrolysis Method.“, 2015.
- [6] Xiaoyuan Wang, Xiaotao Min, Pengfei Zhou, Dongsheng Yu, "Hyperchaotic Circuit Based on Memristor Feedback with Multistability and Symmetries", 2020.
- [7] L. Chua, „Memristor-The Missing Circuit Element“, 2015.
- [8] Pershin, Yuriy & Di Ventra, Massimiliano, „Practical Approach to Programmable Analog Circuits With Memristors.“ 2010.
- [9] G. M. Huang, Y. Ho and P. Li, "Memristor system properties and its design applications to circuits such as nonvolatile memristor memories," 2010 International Conference on Communications, Circuits and Systems (ICCCAS), 2010, pp. 805-810, doi: 10.1109/ICCCAS.2010.5581867.

SAŽETAK

U ovom završnom radu opisan je memristor i njegova svojstva. Prikazana je simulacija i način rada dostupnih SPICE modela memristora u programu LTSPICE te mogućnost primjene u analognim sklopovima. Zbog svojstva pamćenja memristor ima veliki potencijal u primjeni izrade memorije. Ovim radom je opisano nekoliko modela koji su potkrepljeni simulacijom u razvojnom okruženju LTspice. Prilikom simulacije korišteno je osam SPICE modela memristora, a to su: Strukov, Biolek, Joglekar, Knowm, Pino, Michigan, Yakopcic i izrađeni IM model.

Ključne riječi: memristor, SPICE model, strujno-naponska karakteristika, simulacija.

ABSTRACT

This final paper describes the memristor and its properties. The simulation and mode of operation of the available SPICE models of memristors in the LTSPICE program and the possibility of application in analog circuits are presented. Due to its memory properties, the memristor has great potential in the application of memory fabrication. This paper describes several models that are supported by simulation in the LTspice development environment. During the simulation, eight SPICE models of memristors were used, namely: Strukov, Biolek, Joglekar, Knowm, Pino, Michigan, Yakopcic and the developed IM model.

Keywords: memristor, SPICE model, current-voltage characteristic, simulation.

ŽIVOTOPIS

Ivan Marinić rođen je 03. kolovoza 1999. godine u Slavonskom Brodu. Pohađao je Osnovnu školu „Bogoslav Šulek“ u Slavonskom Brodu. Nakon završene osnovne škole, upisao je Tehničku školu Slavonski Brod. Nakon završetka srednje škole, 2018. godine upisuje preddiplomski sveučilišni studij Elektrotehnike na FERIT-u, gdje se opredjeljuje za izborni blok „Komunikacije i informatika“.

Ivan Marinić

PRILOZI

IVAN MARINIĆ MEMRISTOR

```
.subckt IM te be Y
```

```
.params ron=100 roff=16e3 k=10e3 b=10.23 cc=2.11 Cint=1 uthr1=-0.1 uthr2=0.1 mm=0.0001
```

```
Cint Y 0 1 IC=0.3
```

```
Rad Y 0 1G
```

```
Gy 0 Y value={ (k*V(te,be)*(1/(ron*(V(Y))+roff*(1-V(Y))))*(1-(pow((2*V(Y)-1),(2*round(b/cc+abs(V(te,be))))))))*((stpp((uthr1-V(te,be)),mm))+(stpp((V(te,be)-uthr2),mm)))}
```

```
G1 te be value={ V(te,be)*((1/(ron*(V(Y))+roff*(1-V(Y)))))}
```

```
.func stpp(x,p)={0.5*(1+(x/sqrt(pow(x,2)+p)))}
```

```
.ends IM
```

STRUKOV MEMRISTOR

*Titanium dioxide memristor model with Strukov-Williams's window function

* $f_{sw}=x*(1-x)$, where x is the memristor state variable

*representing the nonlinear ionic dopant drift for high-level signals

*terminals: top electrode (te), bottom electrode (be) and electrode for

*measuring the state variable (Y)

*The model is appropriate for low-frequency and high-amplitude signals

*for analysis of titanium dioxide based memristors and memristor circuits

.subckt MEM_STRUKOV te be Y

.params ron=100 roff=16e3 k=10e3 Cint=1

*capacitor for integration the state variable time derivative

Cint Y 0 1 IC=0.3

*additional high-valued resistance for preventing convergence problems

Rad Y 0 1G

*memristor state modeling

Gy 0 Y value={ $(k*V(te,be)*(1/(ron*(V(Y))+roff*(1-V(Y))))*(4*V(Y)*(1-V(Y))))$ }

*memristor conductance

G1 te be value={ $V(te,be)*((1/(ron*(V(Y))+roff*(1-V(Y))))$ }

.ends MEM_STRUKOV

*properties: low operating frequencies, low and middle signal levels, middle nonlinearity,

*low accuracy, activation thresholds - not used, soft-switching operating mode, partial

*Sources: 1. Strukov, D.; Snider, G.; Stewart, D.; Williams, R.S. The Missing Memristor Found. Nat. Lett. 2008, 453, doi:10.1038/nature06932, pp. 80–83.

* 2. Joglekar, Y.; Wolf, S. The Elusive Memristor: Properties of Basic

*Electrical Circuits. Eur. J. Phys. 2009, Vol. 30, pp. 661–675.

*3. Biolek, Z.; Biolek, D.; Biolkova, V. SPICE Model of Memristor with Nonlinear

*Dopant Drift. Radioengineering 2009, 18, pp. 210–214.

BIOLEK MEMRISTOR

```
.SUBCKT MEM_BIOLEK TE BE XSV
```

```
.params Ron=100 Roff=1K x0=.076 D=16N uv=40F p=7
```

```
* Biolek Window Function
```

```
.func f(V1,I1)={1-pow((V1-stp(-I1)),(2*p))}
```

```
* Memristor I-V Relationship
```

```
.func IVRel(V1,V2) = V1/(Ron*V2 + Roff*(1-V2))
```

```
* Circuit to determine state variable
```

```
Gx 0 XSV value={I(Gmem)*Ron*uv*f(V(XSV,0),I(Gmem))/pow(D,2)}
```

```
Cx XSV 0 {1}
```

```
.ic V(XSV) = x0
```

```
* Current source representing memristor
```

```
Gmem TE BE value={IVRel(V(TE,BE),V(XSV,0))}
```

```
.ENDS MEM_BIOLEK
```

*Source: 1. Biolek, Z.; Biolek, D.; Biolkova, V. SPICE Model of Memristor with Nonlinear

*Dopant Drift. Radioengineering 2009, 18, pp. 210–214.

*2. Analog Devices Corporation. LTspice [Online]. Available at:
<https://www.analog.com/en/index.html#LTspice>

JOGLEKAR MEMRISTOR

```
.SUBCKT MEM_JOGLEKAR TE BE XSV
```

```
.params Ron=100 Roff=10K x0=.56 D=12N uv=50F p=7
```

```
* Joglekar Window Function
```

```
.func f(V1) = 1-pow((2*V1-1),(2*p))
```

```
* Memristor I-V Relationship
```

```
.func IVRel(V1,V2) = V1/(Ron*V2 + Roff*(1-V2))
```

```
* Circuit to determine state variable
```

```
Gx 0 XSV value={ I(Gmem)*Ron*uv*f(V(XSV,0))/pow(D,2) }
```

```
Cx XSV 0 {1}
```

```
.ic V(XSV) = x0
```

```
* Current source representing memristor
```

```
Gmem TE BE value={IVRel(V(TE,BE),V(XSV,0))}
```

```
.ENDS MEM_JOGLEKAR
```

*Sources: 1. Joglekar, Y.; Wolf, S. The Elusive Memristor: Properties of Basic

*Electrical Circuits. Eur. J. Phys. 2009, Vol. 30, pp. 661–675.

*2. Biolek, Z.; Biolek, D.; Biolkova, V. SPICE Model of Memristor with Nonlinear

*Dopant Drift. Radioengineering 2009, 18, pp. 210–214.

KNOWM MEMRISTOR

* Known Mean Metastable Switch Memristor SPICE Model

.SUBCKT MEM_KNOWM TE BE XSV

.params Ron=500 Roff=1500 Voff=0.27 Von=0.27 TAU=0.0001 T=298.5 x0=0

* Function G(V(t)) - Describes the device threshold

.func G(V) = V/Ron+(1-V)/Roff

* Function F(V(t),x(t)) - Describes the SV motion

.func F(V1,V2) = (1/TAU)*((1/(1+exp(-1/(T*boltz/echarge)*(V1-Von))))*(1-V2)-(1/(1+exp(-1/(T*boltz/echarge)*(V1+Voff)))))*V2

* Memristor I-V Relationship

.func IVRel(V1,V2) = V1*G(V2)

* Circuit to determine state variable

* $dx/dt = F(V(t),x(t))*G(V(t))$

Cx XSV 0 {1}

.ic V(XSV) = x0

Gx 0 XSV value={F(V(TE,BE),V(XSV,0))}

* Current source for memristor IV response

Gmem TE BE value={IVRel(V(TE,BE),V(XSV,0))}

.ENDS MEM_KNOWM

* Copyright Tim Molter Knowm Inc. 2017.

MICHIGAN MEMRISTOR

* Memristor subcircuit developed by Chang et al.

* Connections:

* TE: Top electrode

* BE: Bottom electrode

* XSV: External connection to plot state variable

```
.SUBCKT MEM_UMICH TE BE XSV
```

```
.param alpha=0.5e-6 beta=0.5 gamma=4e-6 delta=2 xmax=1 xmin=0 drift_bit = 0 lambda=4.5  
eta1=0.004 eta2=4 tau=10
```

```
.param cp={1}
```

```
Cpvar XSV 0 {cp}
```

* Rate equation for state variable

```
Gx 0 XSV value={ trunc(V(TE,BE),cp*V(XSV))*lambda*(eta1*sinh(eta2*V(TE,BE))-  
drift_bit*cp*V(XSV)/tau) }
```

```
.ic V(XSV) = 0.0
```

* Auxiliary functions to limit the range of x

```
.func sign2(var) {(sgn(var)+1)/2}
```

```
.func trunc(var1,var2) {sign2(var1)*sign2(xmax-var2)+sign2(-var1)*sign2(var2-xmin)}
```

* Memristor IV Relationship

```
Gm TE BE value={{(1-cp*V(XSV))*alpha*(1-exp(-  
beta*V(TE,BE)))+(cp*V(XSV))*gamma*sinh(delta*V(TE,BE))}
```

```
.ENDS MEM_UMICH
```

PINO MEMRISTOR

* Code for memristor model proposed by Dr. Pino et al.

* Connections:

* TE: Top electrode

* BE: Bottom electrode

* RSV: External connection to plot resistance

* that is not used otherwise

.SUBCKT MEM_PINO TE BE RSV

.params Ron=160 Roff=1200 Th=0.2 Tl=-0.35 Kh1=5.5e6 Kh2=-20 K11=4e6 K12=20

* Fits the change in resistance to characterization data

.func Rt(V1, V2) = IF(V1 <= Th, IF(V1 >= Tl, 0, IF(V2 < Roff, K11*exp(K12*(V1-Tl)), 0)),
IF(V2 > Ron, - Kh1*exp(Kh2*(V1-Th)), 0))

* Circuit to integrate to find resistance

Gx 0 RSV value={Rt(V(TE,BE),V(RSV))}

Cx RSV 0 {1}

.ic V(RSV) = Roff

* Current source representing memristor

Gmem TE BE value = {V(TE,BE)/V(RSV)}

.ENDS MEM_PINO

YAKOPCIC MEMRISTOR

.subckt MEM_YAKOPCIC TE BE XSV

*.params a1=0.17 a2=0.17 b=0.05 Vp=0.16 Vn=0.15

*+Ap=4000 An=4000 xp=0.3 xn=0.5 alphap=1 alphan=5

*+xo=0.11 eta=1

.params a1=.17 a2=.17 b=0.05 Vp=.65 Vn=0.56 Ap=4000 An=4000 xp=0.3 xn=0.5 alphap=1
alphan=5 xo=0.11 eta=1

* Multiplicative functions to ensure zero state

* variable motion at memristor boundaries

.func wp(V) = xp/(1-xp) - V/(1-xp) + 1

.func wn(V) = V/(1-xn)

* Function G(V(t)) - Describes the device threshold

.func G(V) = IF(V <= Vp, IF(V >= -Vn, 0, -An*(exp(-V)-exp(Vn))), Ap*(exp(V)-exp(Vp)))

* Function F(V(t),x(t)) - Describes the SV motion

.func F(V1,V2) = IF(eta*V1 >= 0, IF(V2 >= xp, exp(-alphap*(V2-xp))*wp(V2), 1), IF(V2 <= (1-xn), exp(alphan*(V2+xn-1))*wn(V2), 1))

* IV Response - Hyperbolic sine due to MIM structure

.func IVRel(V1,V2) = IF(V1 >= 0, a1*V2*sinh(b*V1), a2*V2*sinh(b*V1))

* Circuit to determine state variable

* dx/dt = F(V(t),x(t))*G(V(t))

Cx XSV 0 {1}

.ic V(XSV) = xo

Gx 0 XSV value={eta*F(V(TE,BE),V(XSV,0))*G(V(TE,BE))}

* Current source for memristor IV response

Gm TE BE value = {IVRel(V(TE,BE),V(XSV,0))}

.ends MEM_YAKOPCIC

* SPICE model for memristive devices

* Created by Chris Yakopcic